




# **Shutter Staking Audit**

Côme du Crest

2024-09-01



## Table of contents

- Table of contents
- Shutter Staking Audit
  - Scope
  - Context
  - Status
- Issues
  - [Low] Keyper can have delegated stake below min value
  - [Info] RewardsDistributor may fail to distribute rewards
  - [Info] Cannot unstake delegatee stakes without timing restriction when delegatee is not keeper
  - [Info] No gap in upgradeable contract storage

## Shutter Staking Audit

This document presents the finding of a smart contract audit conducted by Côme du Crest for Shutter.

### Scope

The scope includes all contracts within `blockful-io/shutter-staking/src` as of commit [a5210f4](#).

### Context

The core of the contracts comprise two staking contracts for regular and delegated staking of SHU tokens as well as a reward distributor contract that distribute SHU tokens to stakers accrued every seconds.

### Status

The report has been sent to the core developers.

## Issues

### [Low] Keyper can have delegated stake below min value

#### Summary

The contract `Staking` enforces a min value for the first stake of a keyper. However, anyone can use `DelegateStaking` to delegate any value to a keyper holding no stake at all. In an abstract way keepers can be responsible for a stake below min value.

#### Vulnerability Detail

`Staking.stake()` enforces a min value for the first stake of a keyper:

```
1  function stake(  
2      uint256 amount  
3  ) external updateRewards returns (uint256 stakeId) {  
4      require(keypers[msg.sender], OnlyKeyper());  
5  
6      require(amount > 0, ZeroAmount());  
7  
8      // Get the keyper stakes  
9      EnumerableSetLib.Uint256Set storage stakesIds = userStakes[msg.sender];  
10  
11     // If the keyper has no stakes, the first stake must be at least the  
12     // minimum stake  
13     if (stakesIds.length() == 0) {  
14         require(amount >= minStake, FirstStakeLessThanMinStake());  
15     }  
16     stakeId = nextStakeId++;  
17  
18     // Add the stake id to the user stakes  
19     userStakes[msg.sender].add(stakeId);  
20  
21     // Add the stake to the stakes mapping  
22     stakes[stakeId].amount = amount;  
23     stakes[stakeId].timestamp = block.timestamp;  
24     stakes[stakeId].lockPeriod = lockPeriod;  
25  
26     _deposit(amount);  
27  
28     emit Staked(msg.sender, amount, lockPeriod);  
29 }
```

Anyone can delegate via `DelegateStaking.stake()` to a keyper with no restriction on the stake amount:

```
1  function stake(  
2      uint256 amount  
3  ) external updateRewards returns (uint256 stakeId) {  
4      require(keypers[msg.sender], OnlyKeyper());  
5  
6      require(amount > 0, ZeroAmount());  
7  
8      // Get the keyper stakes  
9      EnumerableSetLib.Uint256Set storage stakesIds = userStakes[msg.sender];  
10  
11     // If the keyper has no stakes, the first stake must be at least the  
12     // minimum stake  
13     if (stakesIds.length() == 0) {  
14         require(amount >= minStake, FirstStakeLessThanMinStake());  
15     }  
16     stakeId = nextStakeId++;  
17  
18     // Add the stake id to the user stakes  
19     userStakes[msg.sender].add(stakeId);  
20  
21     // Add the stake to the stakes mapping  
22     stakes[stakeId].amount = amount;  
23     stakes[stakeId].timestamp = block.timestamp;  
24     stakes[stakeId].lockPeriod = lockPeriod;  
25  
26     _deposit(amount);  
27  
28     emit Staked(msg.sender, amount, lockPeriod);  
29 }
```

```
2         address keyper,
3         uint256 amount
4     ) external updateRewards returns (uint256 stakeId) {
5         require(amount > 0, ZeroAmount());
6
7         require(staking.keypers(keyper), AddressIsNotAKeyper());
8
9         stakeId = nextStakeId++;
10
11         // Add the stake id to the user stakes
12         userStakes[msg.sender].add(stakeId);
13
14         // Add the stake to the stakes mapping
15         stakes[stakeId].keyper = keyper;
16         stakes[stakeId].amount = amount;
17         stakes[stakeId].timestamp = block.timestamp;
18         stakes[stakeId].lockPeriod = lockPeriod;
19
20         // Increase the keyper total delegated amount
21         unchecked {
22             totalDelegated[keyper] += amount;
23         }
24
25         _deposit(amount);
26
27         emit Staked(msg.sender, keyper, amount, lockPeriod);
28     }
```

## Impact

If we take into account the self-locked stake and delegated stake, a keyper could have a stake below `Staking.minStake`.

## Code Snippets

<https://github.com/blockful-io/shutter-staking/blob/a5210f40d61fc6f002b0ed48e46a327aa56975f4/src/Staking.sol#L145-L173>

<https://github.com/blockful-io/shutter-staking/blob/a5210f40d61fc6f002b0ed48e46a327aa56975f4/src/DelegateStaking.sol#L144-L171>

## Recommendation

In `DelegateStaking.stake()` enforce that the delegatee has at least `minStake` locked.

## [Info] RewardsDistributor may fail to distribute rewards

### Summary

When rewards are not distributed for a long period of time, the total reward to distribute may accumulate above the balance of the `RewardsDistributor` which will fail to distribute any rewards.

### Vulnerability Detail

The function to distribute rewards `collectRewards()` will return 0 when rewards exceed the balance of the contract:

```
1  function collectRewards() external override returns (uint256 rewards) {
2      RewardConfiguration storage rewardConfiguration = rewardConfigurations[
3          msg.sender
4      ];
5
6      // difference in time since last update
7      uint256 timeDelta = block.timestamp - rewardConfiguration.lastUpdate;
8
9      rewards = rewardConfiguration.emissionRate * timeDelta;
10
11     // the contract must have enough funds to distribute
12     // we don't want to revert in case its zero to not block the staking
13     // contract
14     if (rewards == 0 || rewardToken.balanceOf(address(this)) < rewards) {
15         return 0;
16     } // @audit if rewards are not collected for a while, they will
17     // accumulate past the balance of the contract and not be distributed
18
19     // update the last update timestamp
20     rewardConfiguration.lastUpdate = block.timestamp;
21
22     // transfer the reward
23     rewardToken.transfer(msg.sender, rewards);
24     emit RewardCollected(msg.sender, rewards);
25 }
```

### Impact

I do not see a strong incentive to collect rewards regularly. As such, we could reach a point where the `RewardsDistributor` owes more rewards than its balance and is unable to disburse them.

The only solution to distribute rewards correctly if that happens is via the admin function `withdrawFunds()` to withdraw funds to the `Staking` contract.

**Code Snippets**

<https://github.com/blockful-io/shutter-staking/blob/a5210f40d61fc6f002b0ed48e46a327aa56975f4/src/RewardsDistributor.sol#L89-L91>

**Recommendation**

Acknowledge that the `RewardsDistributor` contract will always hold enough funds to collect rewards or adapt the function so that it distribute as much rewards as possible and update the timestamp proportionally to the amount of distributed reward. This will have the drawback that rewards may be distributed unfairly to one collector above other collectors.

## [Info] Cannot unstake delegatee stakes without timing restriction when delegatee is not keeper

### Summary

For regular staking, unstaking is allowed with no timing and min stake restriction when staker is no longer a keeper. This is not the case for delegated staking which remains locked for the whole duration even when delegatee is no longer a keeper.

### Vulnerability Detail

In `Staking.unstake()` if the `keyper` argument is no longer registered as a keeper, the stake can be withdrawn before the end of the `lockPeriod`:

```
1     function unstake(  
2         address keyper,  
3         uint256 stakeId,  
4         uint256 _amount  
5     ) external updateRewards returns (uint256 amount) {  
6         ...  
7  
8         // Checks below only apply if keyper is still a keeper  
9         // if keyper is not a keeper anymore, anyone can unstake for them, lock  
10        period is  
11        // ignored and minStake is not enforced  
12        if (keepers[keyper]) {  
13            // Only the keyper can unstake  
14            require(msg.sender == keyper, OnlyKeyper());  
15  
16            ...  
17            uint256 lock = keyperStake.lockPeriod > lockPeriod  
18                ? lockPeriod  
19                : keyperStake.lockPeriod;  
20  
21            unchecked {  
22                require(  
23                    block.timestamp > keyperStake.timestamp + lock,  
24                    StakeIsStillLocked()  
25                );  
26            }  
27  
28            ...  
29            uint256 maxWithdrawAvailable = convertToAssets(balanceOf(keyper)) -  
30                minStake;  
31            require(amount <= maxWithdrawAvailable, WithdrawAmountTooHigh());  
32        }  
33  
34        ...  
35        uint256 shares = _withdraw(keyper, amount);
```



```
36
37     emit Unstaked(keyper, amount, shares);
38 }
```

On the contrary `DelegateStaking.unstake()` enforces the `lockPeriod` no matter the status of the keyper:

```
1  function unstake(
2      uint256 stakeId,
3      uint256 _amount
4  ) external updateRewards returns (uint256 amount) {
5      require(
6          userStakes[msg.sender].contains(stakeId),
7          StakeDoesNotBelongToUser()
8      );
9      Stake memory userStake = stakes[stakeId];
10
11      ...
12      uint256 lock = userStake.lockPeriod > lockPeriod
13          ? lockPeriod
14          : userStake.lockPeriod;
15
16      unchecked {
17          require(
18              block.timestamp > userStake.timestamp + lock,
19              StakeIsStillLocked()
20          );
21
22          // Decrease the amount from the stake
23          stakes[stakeId].amount -= amount;
24
25          // Decrease the total delegated amount
26          totalDelegated[userStake.keyper] -= amount;
27      }
28      ...
29 }
```

## Impact

Discrepancy in between direct and delegated staking. If possible, users are better off staking using a vault contract that redistribute stake rewards and is set as a keyper than to use the delegated staking system implemented by Shutter.

## Code Snippets

<https://github.com/blockful-io/shutter-staking/blob/a5210f40d61fc6f002b0ed48e46a327aa56975f4/src/Staking.sol#L209-L237>

<https://github.com/blockful-io/shutter-staking/blob/a5210f40d61fc6f002b0ed48e46a327aa56975f4/src/DelegateStaking.sol#L188-L235>

**Recommendation**

Allow to unstake immediately in `De`[legateStaking](#) when delegatee is no longer a keyper.

## [Info] No gap in upgradeable contract storage

### Summary

The `BaseStaking` contract provision for the upgradeable proxy pattern but does not declare a `__gap` value for storage as common for these type of contracts in case storage values need to be added to the implementation.

### Vulnerability Detail

The `BaseStaking` contract inherits from upgradeable versions of Openzeppelin contracts and indicate wanting to implement an upgradeable proxy pattern but does not declare a `__gap` storage value to provision for future storage use:

```
1  import {ERC20VotesUpgradeable as ERC20Votes} from "@openzeppelin-upgradeable/
   contracts/token/ERC20/extensions/ERC20VotesUpgradeable.sol";
2
3  abstract contract BaseStaking is OwnableUpgradeable, ERC20Votes {
4      ...
5      constructor() {
6          _disableInitializers();
7      }
8      ...
9  }
```

Both `Staking` and `DelegateStaking` inherit from `BaseStaking` and use storage:

```
1  contract Staking is BaseStaking {
2      ...
3  }
4
5  contract DelegateStaking is BaseStaking {
6      ...
7  }
```

### Impact

If storage values are added in `BaseStaking` in future version of the contracts, there will be a conflict of storage layout with unpredictable impact.

### Code Snippets

<https://github.com/blockful-io/shutter-staking/blob/a5210f40d61fc6f002b0ed48e46a327aa56975f4/src/BaseStaking.sol#L11-L25>

**Recommendation**

Follow Openzeppelin's recommendation of declaring a `uint256[47] __gap` value where 47 is 50 minus the three storage slots already used by `BaseStaking`. See documentation.